

A Punk Meta-Font

Donald E. Knuth
Stanford University

In February, 1985, Gerard and Marjan Unger gave a series of nine evening lectures at Stanford, in which they surveyed the evolution of styles in art, architecture, clothing, product design, and typography during the past 75 years. The lectures were especially interesting because they revealed the way in which changes in typographic fashions were juxtaposed with the changes in other kinds of fashions. The Ungers demonstrated a remarkable fact, that *typography tends to lag behind other stylistic changes by about ten years*.

When I woke up on the morning of their final lecture, I suddenly realized that there was an obvious corollary of what they had been saying during the previous eight evenings: It was now about time to design a typeface based on trends that had emerged during the late 70s! Furthermore, I also had a reasonably clear idea of what such a design might be like, because the lectures had turned up a strong similarity between some "punk" graphics exhibited in London and a certain lines-and-dots motif found in the upholstery of some "punk" furniture designed in Italy.

A lines-and-dots motif is trivially easy for METAFONT to handle, so I decided to create a new family of typefaces called PUNK. I spent several pleasant hours at the computer terminal that afternoon; and by evening I was able to present everybody in the audience with an up-to-the-minute souvenir of the Unger's lectures, laserprinted in PUNK48 and PUNK28.

The idea of PUNK was to start with more-or-less traditional stick-letter shapes, but to ask METAFONT to perturb the key points by random amounts so that the letters look a bit deranged. Here, for example, are several texts set with a few varieties of PUNK fonts:

ABCDEF GHIJKLMNOPQRSTUVWXYZ ABCDEF GHIJKLMNOPQRSTUVWXYZ 0123456789 AEOI ΔΘΛΞ
ΠΣΤΦΨΩ AEOI ΔΘΛΞ ΠΣΤΦΨΩ . . . ; : ? ! ' " # \$ % & * ~ - * () [] ↑ ↓ + / < = >

PUNK10

'HUMPH!' SAID ARTIE. HIS FACE WAS RED AND HE WAS CERTAINLY FLUSTERED. 'IT 'D
BE A DEAD LUCKY THING IF SOME MORE PEOPLE AROUND THE SHOP 'D CHANGE A LITTLE.
THEY COULD N'T BE ANY FUNKER 'N THEY ARE NOW.' [1]

ABCDEF GHIJKLMNOPQRSTUVWXYZ ABCDEF GHIJKLMNOPQRSTUVWXYZ 01234567
89 AEOI ΔΘΛΞ ΠΣΤΦΨΩ AEOI ΔΘΛΞ ΠΣΤΦΨΩ . . . ; : ? ! ' " # \$ % & * ~ - * () []
↑ ↓ + / < = >

PUNK12

A FORKED VEIN BEGAN TO SWELL IN SPADE'S FOREHEAD. ... HIS VOICE
BECAME PERSUASIVE AGAIN. 'LISTEN, GUTMAN, WE'VE ABSOLUTELY GOT
TO GIVE THEM A VICTIM. THERE'S NO WAY OUT OF IT. (LET'S GIVE THEM
THE PUNK.' HE NODDED PLEASANTLY AT THE BOY IN THE DOORWAY. [3]

ABCDEF GHIJKLMNOPQRSTUVWXYZ ABCDEF GHIJKLMNOP
OPQRSTUVWXYZ 0123456789 AEOI ΔΘΛΞ ΠΣΤΦΨΩ AEOI
SIJ ' " # \$ % & * ~ - * () []
↑ ↓ + / < = >

PUNK20

PISTOL. [ASIDE.] THIS PUNK IS ONE OF CUPID'S
CARRIERS. CLAP ON MORE SAILS, PURSUE; UP

WITH YOUR FIGHTS; GIVE FIRE! SHE IS MY PRIZE,
OR OCEAN WHELM THEM ALL! [EXIT.] [6]

ABCDEFGHIJKLMNOPQRSTUVWXYZ ABCDEFGHIJKLMNOP
OPQRSTUVWXYZ 0123456789 AÆØ Γ Δ Θ Λ Ξ Π Σ Υ ϒ Ψ Ω Æ G Ø ß I J
K L M N O P Q R S T U V W X Y Z [] { } ~ ! " # \$ % & ' () * + , - . / : ;
< = > ? @ [] ^ _ ` { } | ~ ! " # \$ % & ' () * + , - . / : ; < = > ? @

PUNKSL20

PUNK ROCK IS THE GENERIC TERM FOR THE LATEST
MUSICAL GARBAGE BRED BY OUR TROUBLED CUL-
TURE, BRITISH AND AMERICAN.

JOHNNY ROTTEN AND THE SEX PISTOLS ARE PUNKS.
THEY SING 'ANARCHY IN THE UK,' WHICH ENDS
WITH A SCREAM: 'DESTROY.' GLASH AND DAMNED
ARE OTHER BANDS.

PUNK WILL FADE. ITS APOLOGISTS ARE
LUDICROUS. THERE ARE WAYS TO PRO-
TEST ABOUT THE PUTRID FACES OF BOTH
POP AND SOCIETY WITHOUT RELAPSING
INTO BARBARISM. PUNK IS ANTI-LIFE,
ANTI-HUMANITY. [2]


ABCDEFGHIJKLMNOPQRSTUVWXYZ ABCDEF
GHIJKLMNOPQRSTUVWXYZ 0123456789 AÆØ
Γ Δ Θ Λ Ξ Π Σ Υ ϒ Ψ Ω Æ G Ø ß I J
K L M N O P Q R S T U V W X Y Z [] { } ~ ! " # \$ % & ' () * + , - . / : ;
< = > ? @ [] ^ _ ` { } | ~ ! " # \$ % & ' () * + , - . / : ; < = > ? @

PUNKBX20

GENERATION OF LETTERFORMS BY MATH-
EMATICAL MEANS WAS FIRST TRIED IN
THE FIFTEENTH CENTURY; IT BECAME

POPULAR IN THE SIXTEENTH AND SEVENTEENTH CENTURIES; AND IT WAS ABANDONED (FOR GOOD REASONS) DURING THE EIGHTEENTH CENTURY. [4]


At the time I "designed" these typefaces, I had just begun to make the final version of the Computer Modern fonts by converting my Almost Computer Modern code to the conventions of METAFONT 84. The letters 'A', 'B', 'C', and 'D' had been debugged so far, and I was planning to tackle 'E' soon; but I felt like taking a break. So I made sketches of some punkish forms, as follows:



(I used a large sheet of graph paper; this illustration has been reduced to about 1/4 of the original size.)

At 1 p.m. I went to the computer and began to compose a simple base file. Not much had to be done, since plain METAFONT already includes most of the basic routines; so I had my first proof output at 1:20 p.m.:

METAFONT output 1985.02.28.13:20 Page 1 (3x3cm) 88




(40% of original size)

The letter 'A' seemed to be working, so I proceeded to type the METAFONT programs for 'B' through 'Z'. I decided to type everything before looking at any proofs, so I simply translated the sketches into METAFONT constructions, composing everything at the keyboard. It wasn't necessary to make accurate measurements, because random perturbations to the points were going to be made anyway; so I soon got used to the

conventions of this font, and I was limited only by typing speed. I didn't even need sketches of the letters 'V' through 'Z', because it was easy to imagine what they would be in algebraic form. At 3:04 p.m. the typing was done, and I was able to run METAFONT and get proofs of all 26 uppercase letters. I also had thrown in a few punctuation marks (period, comma, opening and closing single and double quotation marks).

Of course there were bugs in my code. For example, the first few letters came out looking like this:

METAFONT output 1985.02.28:1504




(I refuse to show you the first form of the letter 'G'.) But by 4 p.m. I was ready to make the first trial setting of text:

THIS 'PUNK' ALPHABET, INSPIRED BY MARJAN'S LECTURE LAST
NIGHT, WAS DESIGNED BY METAFONT'S RANDOM NUMBER GENER-
ATOR.
THE QUICK BROWN FOX JUMPED OVER THE LAZY HAMBURGER FON.

At this point an unexpected glitch slowed things down a bit: The letters of this font had some unusual characteristics that hadn't arisen in GF files before, so a bug showed up in our METAFONT-to-laserprinter software. I made copies of the offending files, for later reference, and I was able to get around the bug by choosing another random seed and generating the font again. After another half hour of tuning things up (and toning down the randomness a bit), I was able to go home for supper.

During the supper hour, I realized that a proper keepsake for that evening would include the typeset date. So I gulped down my meal, quickly sketched a set of numerals, and raced back to my office. Soon I had the font of 43 characters shown at the top of the next page. Whew! I was ready to hand out a sample sheet to everybody at 7 p.m., hot off the copy machine.

A year or so later, I was wondering around in Boston's Museum of Fine Arts and I came across a drawing made by Picasso in 1924 [5]. (See the illustration at the right.) This made me wonder if the PUNK fonts weren't really sixty years behind the times, not just ten. On the other hand, I found a striking confirmation of the relevance of at least part of the PUNK design in October, 1986, when I chanced to see the following typography on a billboard in the Paris Métro(!):



Test of punk20 on February 28, 1985 at 1824

	'0	'1	'2	'3	'4	'5	'6	'7	
'04x			ʀ					ʁ	"2x
'05x						ˆ	˙		
'06x	0	1	2	3	4	5	6	7	"3x
'07x	8	9							
'10x		A	B	C	D	E	F	G	"4x
'11x	H	I	J	K	L	M	N	O	
'12x	P	Q	R	S	T	U	V	W	"5x
'13x	X	Y	Z						
'14x	?								"6x
'15x									
	"8	"9	"A	"B	"C	"D	"E	"F	

In February, 1987, I decided to extend this original font to the full T_EX character set. The extra programming didn't take long, since I decided to generate the lowercase letters as "small caps," and since each new character could be typed into the computer in one or two minutes. About one third of the characters had to be revised after I saw proofsheets, since they looked either too punk or not punk enough; and one-third of the revised characters had to be revised again; and so on. But after about six hours of additional work, a complete PUNK meta-font with 128 characters was ready for use (in case anybody wanted it).

In the remainder of this paper, I'll present the details of the METAFONT code, since this may be the shortest possible example of METAFONT programs for a family of complete 128-character fonts.

The programs appear in several different kinds of files, as explained in Appendix E of [4]: There are parameter files, to specify specific fonts of the family; there is a driver file, which controls most of the font generation process; and there are program files, which contain the code for individual characters. (I didn't need a base file, since the special macros for these fonts could all be included in the driver file.)

Here is a typical parameter file, PUNK20.MF:

```

% 20-point PUNK font:
designsize := 20pt#; font_identifier := "PUNK";

ht# := 14pt#; % height of characters
u# := 4/9pt#; % unit width
s# := 2pt#; % extra sidebar
px# := .8pt#; % horizontal thickness of pen
py# := .6pt#; % vertical thickness of pen
dot# := 2.7pt#; % diameter of dots
dev# := .5pt#; % standard deviation of punk points
slant := 0; % obliqueness

```

seed := 2.71828;

% seed for random number generator

input PUNK

% switch to the driver file

Its purpose is to customize the meta-design to a particular selection of sizes and weights. The parameters used to define the five fonts exhibited earlier in this article are:

	PUNK10	PUNK12	PUNK20	PUNKSL20	PUNKBX20
<i>designsize</i>	10pt#	12pt#	20pt#	20pt#	20pt#
font_identifier	"PUNK"	"PUNK"	"PUNK"	"PUNKSL"	"PUNKBX"
<i>ht#</i>	7pt#	8.4pt#	14pt#	14pt#	14pt#
<i>u#</i>	1/4pt#	.3pt#	4/9pt#	4/9pt#	.6pt#
<i>s#</i>	1.2pt#	1.4pt#	2pt#	2pt#	2.2pt#
<i>px#</i>	.6pt#	.75pt#	.8pt#	.8pt#	2pt#
<i>py#</i>	.5pt#	.62pt#	.6pt#	.6pt#	1.6pt#
<i>dot#</i>	1.3pt#	1.6pt#	2.7pt#	2.7pt#	3.5pt#
<i>dev#</i>	.3pt#	.36pt#	.5pt#	.5pt#	.5pt#
<i>slant</i>	0	0	0	1/3	0
<i>seed</i>	sqrt 2	sqrt 3	2.71828	3.14159	0.57722

The driver file PUNK.MF was the most difficult to write, because it contains the "essence" of the design. The various parts of this file grew one step at a time. For example, the last two parameters of the '**beginpunkchar**' macro were added after I noticed that some characters can't tolerate as much random deviation in their points as a normal character can (otherwise they become unrecognizable).

% This is PUNK, a meta-font inspired by Gerard and Marjan Unger's lectures, February 1985

mode_setup;

randomseed := *seed*;

define_pixels(*u*, *dev*);

define_blacker_pixels(*px*, *py*, *dot*);

define_whole_pixels(*s*);

xoffset := *s*;

pickup pencircle *xscaled px yscaled py*; *punk_pen* := *savepen*;

pickup pencircle *scaled dot*; *def_pen_path*;

path *dot_pen_path*; *dot_pen_path* := *currentpen_path*;

currenttransform := *identity* *slanted slant yscaled aspect_ratio*;

def beginpunkchar(**expr** *c*, *n*, *h*, *v*) =

% code *c*; width is *n* units

hdev := *h * dev*; *vdev* := *v * dev*;

% modify horizontal and vertical amounts of deviation

beginchar(*c*, *n * u#*, *ht#*, 0); **italcorr** *ht# * slant*; **pickup** *punk_pen* **enddef**;

extra_endchar := *extra_endchar* & "w:=w+2s;charwd:=charwd+2s#";

def ↑ = transformed *currenttransform* **enddef**;

def *makebox*(**text** *rule*) =

for *y* = 0, *h*:

rule((-*s*, *y*)↑, (*w* - *s*, *y*)↑); **endfor**

% horizontals

for *x* = -*s*, 0, *w* - 2*s*, *w* - *s*: *rule*((*x*, 0)↑, (*x*, *h*)↑); **endfor**

% verticals

enddef;

rulepen := *pensquare*;

vardef *pp* **expr** *z* = *z* + (*hdev* * *normaldeviate*, *vdev* * *normaldeviate*) **enddef**;

def *pd* **expr** *z* = *addto_currentpicture* **contour**

dot_pen_path shifted *z_t_* **withpen** *penspeck* **enddef**;

% drawdot

input PUNKL

% uppercase letters

input PUNKAE

% uppercase Æ, Œ, O

input PUNKG

% uppercase greek

```

input PUNKP                                % punctuation
input PUNKD                                % digits
input PUNKA                                % accents
ht# := .6ht#; dev := .7dev;
input PUNKSL                               % special lowercase
extra_beginchar := extra_beginchar & "charcode:=charcode+32;";
input PUNKL                                % lowercase letters
extra_beginchar := extra_beginchar & "charcode:=charcode-35;";
input PUNKAE                               % lowercase æ, œ, ø
font_slant := slant;
font_quad := 18u# + 2s#;
font_normal_space := 9u# + 2s#;
font_normal_stretch := 6u#;
font_normal_shrink := 4u#;
font_x_height := ht#;
font_coding_scheme := "TeX text without f-ligatures";
end
    
```

The 128 characters generated by PUNK.MF have the same font positions as the characters in fonts like cmr5 and cmcsc10 that don't have f-ligatures. Here, for example, is the layout of the font PUNKZ20, which is like PUNK20 except that *dev* = 0 (so that there is no randomness):

	'0	'1	'2	'3	'4	'5	'6	'7	
'00x	Γ	Δ	Θ	Λ	Ξ	Π	Σ	Υ	"0x
'01x	Φ	Ψ	Ω	↑	↓	∴	∵	∶	
'02x	∑	∏	∞	∫	∂	∇	∆	∅	"1x
'03x	∂	∇	∆	∅	∂	∇	∆	∅	
'04x	∂	∇	∆	∅	∂	∇	∆	∅	"2x
'05x	()	∞	∫	∂	∇	∆	∅	
'06x	∅	∂	∇	∆	∅	∂	∇	∆	"3x
'07x	∅	∂	∇	∆	∅	∂	∇	∆	
'10x	Q	A	B	C	D	E	F	G	"4x
'11x	H	I	J	K	L	M	N	O	
'12x	P	Q	R	S	T	U	V	W	"5x
'13x	X	Y	Z	∫	∂	∇	∆	∅	
'14x	∫	A	B	C	D	E	F	G	"6x
'15x	H	I	J	K	L	M	N	O	
'16x	P	Q	R	S	T	U	V	W	"7x
'17x	X	Y	Z	∫	∂	∇	∆	∅	
	"8	"9	"A	"B	"C	"D	"E	"F	

Let's look now at the program files. The first one I wrote was PUNKL.MF, which defines all the letters from \AA to \textbackslash :

```

% Punk letters:
beginpunkchar("A", 13, 1, 2);
z1 = pp(1.5u, 0); z2 = (.5w, 1.1h); z3 = pp(w - 1.5u, 0);
pd z1; pd z3; draw z1 -- z2 -- z3; % left and right diagonals
z4 = pp.3[z1, z2]; z5 = pp.3[z3, z2]; pd z4; pd z5; draw z4 -- z5; % crossbar
endchar;

beginpunkchar("B", 12, 1, 1);
z1 = pp(2u, 0); z2 = pp(2u, .6h); z3 = pp(2u, h); pd z1; pd z3; draw z1 -- z3; % stem
z1.5 = pp(w - u, .5y2); z2.5 = pp(w - u, .5[y2, y3]); draw z2 -- z2.5 -- z3; % upper lobe
draw flex(z2, z1.5, z1); % lower lobe
endchar;

beginpunkchar("C", 13, 1, 2);
z1 = pp(w - 2u, .8h); z2 = pp(.6w, h); z3 = pp(u, .5h); z4 = (.6w, 0); z5 = (w - 2u, .2h);
pd z1; pd z5; draw z1 .. z2 .. z3 .. z4 .. z5; % arc
endchar;

beginpunkchar("D", 14, 1, 2);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(w - u, .6h); pd z1; pd z2; draw flex(z1, z3, z2); % lobe
draw z1 -- z2; % stem
endchar;

beginpunkchar("E", 12, .5, 1);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(w - 2.5u, h); z4 = pp(w - 2u, 0);
pd z3; pd z4; draw z4 -- z1 -- z2 -- z3; % stem and arms
z5 = pp(2u, .6h); z6 = pp(w - 3u, .6h); pd z5; pd z6; draw z5 -- z6; % crossbar
endchar;

beginpunkchar("F", 12, .5, 2);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(w - 2u, h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stem and arm
z5 = pp(2u, .6h); z6 = pp(w - 3u, .6h); z4 = pp.5[z5, z6] - (0, .1h);
pd z5; pd z6; draw flex(z5, z4, z6); % crossbar
endchar;

beginpunkchar("G", 13, .5, .5);
z1 = pp(w - 2u, .8h); z2 = pp(.6w, h); z3 = pp(u, .5h); z4 = pp(.6w, 0); z5 = (w - 2u, 0);
pd z1; draw z1 .. z2 .. z3 .. z4 --- z5; % arc
z6 = pp(.5[u, x5], .4h); pd z6; pd z5; draw z6 -- (pp(x5, y6)) -- z5; % spur
endchar;

beginpunkchar("H", 14, 1, .5);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(w - 2u, 0); z4 = pp(w - 2u, h);
z5 = pp(2u, .6h); z6 = pp(w - 2u, .6h);
pd z1; pd z2; pd z3; pd z4; draw z1 -- z2; draw flex(z3, z6, z4); % stems
pd z5; draw z5 -- z6; % crossbar
endchar;

beginpunkchar("I", 5, 1, 2);
z1 = pp(.5w, 0); z2 = (.5w, 1/3h); z3 = (.5w, 2/3h); z4 = (.5w, h);
pd z1; pd z4; draw flex(z1, z2, z3, z4); % stem
endchar;

beginpunkchar("J", 9, 1, 2);
z1 = pp(w - 2u, h); z2 = pp(w - 2u, -.1h); z3 = pp(u, 0); pd z1; pd z3; draw z1 -- z2 -- z3; % arc
endchar;

```



```

beginpunkchar("K", 14, 1, 2);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(2u, 1/3h); z4 = pp(w - 1.5u, h);
pd z1; pd z2; draw z1 -- z2; % stem
pd z3; pd z4; draw z3 -- z4; % upper diagonal
z6 = pp(w - u, 0); z5 = 1/3[z3, z4]; pd z6; draw flex(z5, .8[z1, 2/3[z5, z6]], z6); % lower diagonal
endchar;

beginpunkchar("L", 11, 1, 2);
z1 = pp(2u, h); z2 = pp(2u, 0); z3 = pp(w - 1.5u, 0);
pd z1; pd z3; draw z1 -- z2 -- z3; % stem and arm
endchar;

beginpunkchar("M", 17, .5, 2);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(.5w, 0); z4 = pp(w - 2u, h); z5 = pp(w - 2u, 0);
pd z1; pd z5; draw z1 -- z2 -- z3 -- z4 -- z5; % stems and diagonals
endchar;

beginpunkchar("N", 13, .75, 2);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(w - 2u, 0); z4 = pp(w - 2u, h);
pd z1; pd z4; draw z1 -- z2 -- z3 -- z4; % stems and diagonals
endchar;

beginpunkchar("O", 12, .5, 2);
z1 = pp(.5w, h); z2 = pp(u, .55h); z3 = pp(.5w, 0); z4 = pp(w - u, .55h);
pd z1; draw z1 {left} .. z2 .. z3 .. z4 .. z1; % bowl
endchar;

beginpunkchar("P", 13, 1, 2);
z1 = pp(2u, 0); z2 = pp(2u, 1.1h); z3 = pp(2u, .5h); z4 = pp(w, .6[y3, y2]);
pd z1; pd z3; draw z1 -- z2 -- z4 -- z3; % stem and bowl
endchar;

beginpunkchar("Q", 14, .5, 2);
z1 = pp(.5w, h); z2 = pp(u, .55h); z3 = pp(.5w, 0); z4 = pp(w - u, .55h);
pd z1; draw z1 {curl 2} .. z2 .. z3 .. z4 .. z1; % bowl
z5 = pp(.4w, .2h); z6 = pp(w - u, -.1h); z7 = pp(.5[x5, x6], -.2h);
pd z5; pd z6; draw z5 -- z7 -- z6; % tail
endchar;

beginpunkchar("R", 16, 1, 2);
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(w - u, .6[y2, y4]); z4 = pp(2u, .5h); z5 = pp(w - 1.5u, 0);
pd z1; pd z2; pd z5; draw z1 -- flex(z2, z3, z4) -- z5; % stem, bowl, and diagonal
endchar;

beginpunkchar("S", 11, .3, 1);
z1 = pp(w - 2u, .9h); z2 = pp(.5w, h); z3 = pp(u, .7h); z4 = .6[z6, z2];
z5 = pp(w - u, .35h); z6 = pp(.5w, u); z7 = pp(u, .2h);
pd z1; pd z7; draw z1 -- z2 ... z3 .. z4 .. z5 ... z6 -- z7; % stroke
endchar;

beginpunkchar("T", 13, .75, 2);
z1 = pp(u, h); z2 = pp(w - u, h); z3 = pp(.5w, 0);
pd z1; pd z2; pd z3; draw z1 -- z2; % arms
draw .5[z1, z2] -- z3; % stem
endchar;

beginpunkchar("U", 13, .3, 2);
z1 = pp(2u, h); z2 = pp(2u, .2h); z3 = pp(.5w, 0); z4 = pp(w - 2u, .2h); z5 = pp(w - 2u, h);
pd z1; pd z5; draw z1 --- z2 ... z3 {z4 - z2} ... z4 --- z5; % stroke
endchar;

```

```

beginpunkchar("V", 13, 1, 2);
z1 = pp(1.5u, h); z2 = pp(.5w, 0); z3 = pp(w - 1.5u, h);
pd z1; pd z3; draw z1 -- z2 -- z3; % diagonals
endchar;

beginpunkchar("W", 18, 1, 2);
z1 = pp(1.5u, h); z2 = pp(.5[x1, x3], 0); z3 = pp(.5w, .8h); z4 = pp(.5[x3, x5], 0); z5 = pp(w - 1.5u, h);
pd z1; pd z5; draw z1 -- z2 -- z3 -- z4 -- z5; % diagonals
endchar;

beginpunkchar("X", 13, 1, 1);
z1 = pp(1.5u, h); z2 = pp(w - 1.5u, 0); z3 = pp(1.5u, 0); z4 = pp(w - 2.5u, h);
pd z1; pd z2; draw z1 -- z2; % main diagonal
pd z3; pd z4; draw z3 -- z4; % cross diagonal
endchar;

beginpunkchar("Y", 13, 1, 2);
z1 = pp(1.5u, h); z2 = pp(w - 1.5u, h); z3 = pp(.5w, .5h); z4 = pp(.5w, 0);
pd z1; pd z2; pd z4; draw z1 -- z3 -- z4; % stem and left diagonal
draw z2 -- z3; % right diagonal
endchar;

beginpunkchar("Z", 11, 1, 2);
z1 = pp(1.5u, h); z2 = pp(w - 2.5u, h); z3 = pp(1.5u, 0); z4 = pp(w - 1.5u, 0);
pd z1; pd z4; draw z1 -- z2 -- z3 -- z4; % diagonals
endchar;

```

(It slowed me down a little to type the comments that identify the strokes. But such comments are enormously valuable when characters are being revised, so I knew that I should include them right from the beginning.)

Three of the letters go into a special file, PUNKAE.MF, because the character codes of these uppercase letters have a nonstandard relation to the character codes of the corresponding lowercase equivalents:

```

beginpunkchar(oct "035", 16, 1, 2); % Æ
z1 = pp(1.5u, 0); z2 = pp(.6w, h); z3 = pp(w - 1.5u, h);
pd z1; pd z3; draw z1 -- z2 -- z3; % left diagonal and upper arm
z4 = pp.3[z1, z2]; z5 = pp(.6w, 0); z6 = pp(w - 2u, .3h); pd z4; pd z6; draw z4 -- z6; % crossbar
z7 = pp(w - u, 0); pd z2; pd z7; draw z2 -- z5 -- z7; % stem and lower arm
endchar;

beginpunkchar(oct "036", 18, 1, 2); % Œ
z1 = pp(.5w, h); z2 = pp(u, .4h); z3 = pp(.5w, 0); pd z1; draw z1 .. z2 .. {right}z3; % bowl
z4 = pp(w - 1.5u, h); z5 = pp(w - 2u, .4h); z6 = pp(w - u, 0);
pd z4; pd z6; draw z4 -- z1 -- z3 -- z6; % arms and stem
pd z5; draw z5 -- .4[z3, z1]; % crossbar
endchar;

beginpunkchar(oct "037", 14, 1, 1); % Ø
z1 = pp(.5w, h); z2 = pp(u, .5h); z3 = pp(.5w, 0); z4 = pp(w - u, .5h);
z5 = pp(w - 2u, 1.1h); z6 = pp(2u, -.1h);
pd z1; pd z6; draw z1 .. z2 .. z3 .. z4 .. z5 -- z6; % bowl and diagonal
endchar;

```

There's also a special file PUNKSL.MF for lowercase letters with no matching uppercase:

```

beginpunkchar(oct "020", 5, 1, 2); % dotless I
z1 = pp(.5w, 0); z2 = (.5w, 1/3h); z3 = (.5w, 2/3h); z4 = (.5w, h);
pd z1; pd z4; draw flex(z1, z2, z3, z4); % stem
endchar;

```

```

beginpunkchar(oct "021", 9, 1, 2); % dotless J
z1 = pp(w - 2u, h); z2 = pp(w - 2u, -.1h); z3 = pp(u, 0); pd z1; pd z3; draw z1 -- z2 -- z3; % arc
endchar;

```

```

beginpunkchar(oct "031", 18, .3, 1); % German sharp S
z1 = pp(.5w - u, .9h); z2 = pp(1/3w, h); z3 = pp(u, .7h); z4 = .6[z6, z2];
z5 = pp(.5w, .35h); z6 = pp(1/3w, u); z7 = pp(u, .2h);
pd z1; pd z7; draw z1 -- z2 ... z3 .. z4 .. z5 ... z6 -- z7; % left stroke
for i = 1 upto 7: z[i + 10] = pp(z[i] shifted (.5w - u, 0)); endfor
pd z11; pd z17; draw z11 -- z12 ... z13 .. z14 .. z15 ... z16 -- z17; % right stroke
endchar;

```

The uppercase Greek letters in file PUNKG.MF may have a slightly different style than those of PUNKL, because I wrote them two years later. Is there an obvious difference?

```

beginpunkchar(oct "000", 11, 1, 2); % Γ
z1 = pp(2u, 0); z2 = pp(2u, h); z3 = pp(w - 1.5u, h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stem and arm
endchar;

```

```

beginpunkchar(oct "001", 15, 1, 2); % Δ
z1 = pp(u, 0); z2 = pp(.5w, h); z3 = pp(w - u, 0);
pd z1; draw z1 -- z2 .. tension 5 .. z3 .. tension 5 .. z1; % triangle
endchar;

```

```

beginpunkchar(oct "002", 15, .5, 2); % Θ
z1 = pp(.5w, h); z2 = pp(u, .6h); z3 = pp(.5w, 0); z4 = pp(w - u, .6h);
pd z1; draw z1 .. tension .8 .. z2 .. z3 .. z4 .. tension .8 .. z1; % bowl
z5 = pp(x2 + 2u, .4h); z6 = pp(x4 - 2u, .4h); pd z5; pd z6; draw z5 -- z6; % bar
endchar;

```

```

beginpunkchar(oct "003", 12, 1, 2); % Λ
z1 = pp(u, 0); z2 = pp(.5w, h); z3 = pp(w - u, 0);
pd z1; pd z3; draw z1 -- z2 -- z3; % diagonals
endchar;

```

```

beginpunkchar(oct "004", 12, 1, 1); % Ξ
z1 = pp(u, h); z2 = pp(w - u, h); pd z1; pd z2; draw z1 -- z2; % upper arm
z3 = pp(2u, .55h); z4 = pp(w - 2u, .55h); pd z3; pd z4; draw z3 -- z4; % bar
z5 = pp(u, 0); z6 = pp(w - u, 0); pd z5; pd z6; draw z5 -- z6; % lower arm
endchar;

```

```

beginpunkchar(oct "005", 13, 1, .5); % Π
z1 = pp(1.5u, 0); z2 = pp(1.5u, h); z3 = pp(w - 1.5u, h); z4 = pp(w - 1.5u, 0);
pd z1; pd z4; draw z1 -- z2 -- z3 -- z4; % stems and bar
endchar;

```

```

beginpunkchar(oct "006", 13, 1, 1); % Σ
z1 = pp(w - u, h); z2 = pp(u, h); z3 = pp(.5w - u, .5h); z4 = pp(u, 0); z5 = pp(w - u, 0);
pd z1; pd z5; draw z1 -- z2{.5[z4, z5] - z2} .. z3 -- z4 -- z5; % arms and diagonals
endchar;

```

```

beginpunkchar(oct "007", 15, 1, .5); % Υ
z1 = pp(u, .8h); z2 = pp(.3w, h); z3 = pp(.5w, .5h); z4 = pp(.5w, 0);
pd z1; pd z4; draw z1 .. z2 .. tension 2 .. z3 --- z4; % left arc and stem
z5 = pp(w - u, .8h); z6 = pp(.7w, h); pd z5; draw z5 .. z6 .. tension 2 .. {z4 - z3}z3; % right arc
endchar;

```

```

beginpunkchar(oct "010", 13, 1, 2); % Φ
z1 = pp(.5w, h); z2 = pp(.5w, 0); pd z1; pd z2; draw z1 -- z2; % stem
z3 = pp(.5w, 2/3h); z4 = pp(u, .5h); z5 = pp(.5w, 1/4h); z6 = pp(w - u, .5h);
pd z3; draw z3 .. z4 .. z5 .. z6 .. z3; % bowl
endchar;

beginpunkchar(oct "011", 14, 1, 1); % Ψ
z1 = pp(.5w, h); z2 = pp(.5w, 0); pd z1; pd z2; draw z1 -- z2; % stem
z3 = pp(u, .8h); z4 = pp(.5w, .2h); z5 = pp(w - u, .8h);
pd z3; pd z5; draw z3{.4[z1, z2] - z3} .. z4{right} .. {z5 - .4[z1, z2]}z5; % stroke
endchar;

beginpunkchar(oct "012", 13, 1, 2); % Ω
z1 = pp(u, 0); z2 = pp(1/3w, 0); z3 = pp(u, 2/3h); z4 = pp(.5w, h);
z5 = pp(w - u, 2/3h); z6 = pp(2/3w, 0); z7 = pp(w - u, 0);
pd z1; pd z7; draw z1 -- z2{up} .. z3 .. z4 .. z5 .. {down}z6 -- z7; % bowl and arms
endchar;

```

The next program file, PUNKD.MF, defines the ten punk digits. I ran out of time while typing this, so the comments at the end are somewhat uninspired:

```

beginpunkchar("0", 9, .5, 1);
z1 = pp(.5w, h); z2 = pp(u, .55h); z3 = pp(.5w, 0); z4 = pp(w - u, .55h);
pd z1; draw z1{curl 2} .. z2 .. z3 .. z4 .. z1; % bowl
endchar;

beginpunkchar("1", 9, .3, 1);
z1 = pp(2u, .7h); z2 = pp(.6w, h); z3 = pp(.6w, 0);
pd z1; pd z3; draw z1 -- z2 -- z3; % serif and stem
endchar;

beginpunkchar("2", 9, 1, 1);
z1 = pp(2u, .7h); z2 = pp(.5w, h); z3 = pp(w - u, .6h); z4 = pp(u, 0); z5 = pp(w - 2u, 0);
pd z1; pd z5; draw z1 .. z2 .. z3 .. z4 -- z5; % stroke
endchar;

beginpunkchar("3", 9, .5, .5);
z1 = pp(2u, .7h); z2 = pp(.5w, h); z3 = pp(w - u, .5[y2, y4]);
z4 = pp(.5w - u, .55h); z5 = pp(w - u, .5[y4, y6]); z6 = pp(.5w, 0); z7 = pp(1.5u, .2h);
pd z1; pd z7; draw z1 .. z2 .. z3 .. z4 & z4 .. z5 .. z6 .. z7; % arcs
endchar;

beginpunkchar("4", 9, 1, 1);
z1 = pp(w - u, .3h); z2 = pp(u, .3h); z3 = pp(2/3w, h); z4 = pp(2/3w, 0);
pd z1; pd z4; draw z1 -- z2 -- z3 -- z4; % stem and diagonals
endchar;

beginpunkchar("5", 9, .5, .5);
z1 = pp(w - 2u, h); z2 = pp(2u, h); z3 = pp(u, .7h); z4 = pp(w - u, .5[y3, y5]);
z5 = pp(.5w, 0); z6 = pp(u, .2h);
pd z1; pd z6; draw z1 -- z2 -- z3 .. z4 .. z5 .. z6; % stroke
endchar;

beginpunkchar("6", 9, 1, 1);
z1 = pp(2/3w, h); z2 = pp(u, .3h); z3 = pp(.5w, 0); z4 = pp(w - u, .3h); z5 = pp(.6w, .6h);
z6 = pp.z 2; pd z1; pd z6; draw z1 .. z2 .. z3 .. z4 .. z5 -- z6; % stroke
endchar;

```

```

beginpunkchar("7", 9, .5, 1);
z1 = pp(2u, h); z2 = pp(w - .5u, h); z3 = pp(.4w, 0);
pd z1; pd z3; draw z1 -- z2 & z2 .. z3{down}; % stroke
endchar;

beginpunkchar("8", 9, .5, .5);
z1 = pp(.5w, h); z2 = pp(u, .5[y1, y3]); z3 = pp(.5w, .6h); z4 = pp(w - u, .5[y3, y5]);
z5 = pp(.5w, 0); z6 = pp(u, .5[y5, y3]); z7 = pp(w - u, .5[y1, y3]);
pd z1; draw z1{curl 8} .. z2 .. z3 .. z4 .. z5 .. z6 .. z3 .. z7 .. z1; % stroke
endchar;

beginpunkchar("9", 9, 1, 1);
z1 = pp(1/3w, 0); z2 = pp(w - u, .7h); z3 = pp(.5w, h); z4 = pp(u, .7h); z5 = pp(.5w, .4h);
pd z1; pd z5; draw z1 .. z2 .. z3 .. z4 .. z5; % stroke
endchar;

```

The program file PUNKP.MF defines "punk punctuation." This was one of the most difficult to write—although most of the characters are very simple—because there are so **DARN** many punctuation marks.

```

beginpunkchar(".", 5, 1, 2);
pd pp(.5w, 0); % dot
endchar;

beginpunkchar(", ", 5, .5, .5);
z1 = pp(.5w, 0); z2 = pp(w - u, -.1h); z3 = pp(.5w, -.3h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stroke
endchar;

beginpunkchar(":", 5, 1, .5);
pd pp(.5w, 0); pd pp(.5w, .4h); % dots
endchar;

beginpunkchar("; ", 5, .5, .5);
z1 = pp(.5w, 0); z2 = pp(w - u, -.1h); z3 = pp(.5w, -.3h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stroke
pd pp(.5w, .4h); % dot
endchar;

beginpunkchar("!", 5, .5, .5);
pd pp(.5w, 0); % dot
z1 = pp(.5w, 1.05h); z2 = pp(.5w, .3h); pd z1; pd z2; draw z1 -- z2; % stem
endchar;

ligtable "!" : " " =: oct "016";

beginpunkchar(oct "016", 5, .5, .5); % Spanish inverted !
pd pp(.5w, .9h); % dot
z1 = pp(.5w, -.1h); z2 = pp(.5w, .6h); pd z1; pd z2; draw z1 -- z2; % stem
endchar;

beginpunkchar("?", 9, 1, .5);
z1 = pp(1.5u, .8h); z2 = pp(.5w, h); z3 = pp(w - u, .8h); z4 = pp(.5w, .3h);
pd z1; pd z4; draw z1 .. z2 .. z3 .. {down}z4; % arc and stem
pd pp(.5w, 0); % dot
endchar;

ligtable "?" : " " =: oct "017";

beginpunkchar(oct "017", 9, 1, .5); % Spanish inverted ?
z1 = pp(1.5u, .1h); z2 = pp(.5w, -.1h); z3 = pp(w - u, .1h); z4 = pp(.5w, .6h);
pd z1; pd z4; draw z1 .. z2 .. z3 .. {up}z4; % arc and stem
pd pp(.5w, .9h); % dot
endchar;

```

```

beginpunkchar("&", 14, .5, .5);
z1 = pp(w - 2u, h); z2 = pp(u, h); z3 = pp(3u, 0); z5 = pp(w - u, .6h); z6 = pp(w - 2u, 0);
pd z1; pd z5; draw z1 -- z2 -- z3 -- z5; % arms and stem
draw z1 -- .5[z2, z3]; pd z6; draw z6 -- .6[z3, z5]; % diagonals
endchar;

beginpunkchar("$", 12, .5, .5);
z1 = pp(w - 1.5u, .7h); z2 = pp(.5w, h); z3 = pp(u, .7h); z4 = .5[z3, z5];
z5 = pp(w - u, .3h); z6 = pp(.5w, 0); z7 = pp(u, .3h);
pd z1; pd z7; draw z1 .. z2 .. z3 .. z4 .. z5 .. z6 .. z7; % stroke
z8 = z2 + (0, .1h); pd z8; draw z8 -- z6; % stem
endchar;

beginpunkchar("%", 18, .5, .5);
z1 = pp(3.5u, 1.1h); z2 = pp(u, .8h); z3 = pp(3.5u, .5h); z4 = pp(6u, .8h);
z5 = pp(w - 3.5u, .5h); z6 = pp(w - 6u, .2h); z7 = pp(w - 3.5u, -.1h); z8 = pp(w - u, .2h);
pd z1; draw z1 .. z2 .. z3 .. z4 .. z1; % upper bowl
pd z5; draw z5 .. z6 .. z7 .. z8 .. z5; % lower bowl
z9 = pp(w - 3u, 1.1h); z0 = pp(3u, -.1h); pd z0; draw z9 -- z0; % diagonal
draw z1{z5 - z1} .. z9; % link
endchar;

beginpunkchar("@", 18, 1, .5);
z1 = pp(2u, 0); z2 = pp(1/3w, .7h); z3 = pp(w - 6u, 0);
z4 = pp(w, .3h); z5 = pp(1/3w, h); z6 = pp(u, .5h); z7 = .7[z2, z3];
pd z1; pd z7; draw z1 -- z2 -- z3{right} .. z4 .. z5 .. z6 .. z7; % diagonals and stroke
endchar;

beginpunkchar("-", 7, .5, .5);
z1 = pp(u, .4h); z2 = pp(w - u, .5h); pd z1; pd z2; draw z1 -- z2; % bar
endchar;
ligtable "-" "-" =: oct "173";

beginpunkchar(oct "173", 9, .5, .5); % -
z1 = pp(0, .5h); z2 = pp(w, .4h); pd z1; pd z2; draw z1 -- z2; % bar
endchar;
ligtable oct "173" "-" =: oct "174";

beginpunkchar(oct "174", 18, .5, .5); % —
z1 = pp(0, .5h); z2 = pp(w, .4h); pd z1; pd z2; draw z1 -- z2; % bar
endchar;

beginpunkchar("+", 9, .5, 1);
z1 = pp(0, .5h); z2 = pp(w, .5h); pd z1; pd z2; draw z1 -- z2; % bar
z3 = pp(.5w, .1h); z4 = pp(.5w, .9h); pd z3; pd z4; draw z3 -- z4; % stem
endchar;

beginpunkchar("*", 13, .5, 1);
z0 = pp(.5w, 1.1h); z1 = pp(u, .9h); z2 = pp(2u, .3h); z3 = pp(w - u, .3h); z4 = pp(w - u, .9h);
pd z0; draw z0 -- z2 .. 1/3[.5[z2, z4], z0] .. z4 -- z1 -- z3 -- z0; % star
endchar;

beginpunkchar("'", 5, .3, .5);
z1 = pp(1.5u, h); z2 = pp(w - u, .85h); z3 = pp(u, 2/3h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stroke
endchar;
ligtable "'" "'" =: oct "042";

```

```

beginpunkchar(oct "042", 9, .3, .5); % "
z1 = pp(.5w - .5u, h); z2 = pp(u, .6h); z3 = pp(w - u, .95h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stroke
endchar;

beginpunkchar("`", 5, .3, .5);
z1 = pp(w - 1.5u, h); z2 = pp(u, .85h); z3 = pp(w - u, 2/3h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stroke
endchar;
ligtable "`": "`" =: oct "134";
beginpunkchar(oct "134", 9, .3, .5); % "
z1 = pp(.5w + .5u, h); z2 = pp(w - u, .6h); z3 = pp(u, .95h);
pd z1; pd z3; draw z1 -- z2 -- z3; % stroke
endchar;

beginpunkchar(oct "015", 9, .3, .5); % '
z1 = pp(.5w, h); z2 = pp(.5w, .6h); pd z1; pd z2; draw z1 -- z2; % stem
endchar;

beginpunkchar("(" , 7, .5, .5);
z1 = pp(w - u, h); z2 = pp(u, .5h); z3 = pp(w - u, 0); pd z1; pd z3; draw z1 .. z2 .. z3; % stroke
endchar;

beginpunkchar(")" , 7, .5, .5);
z1 = pp(u, h); z2 = pp(w - u, .5h); z3 = pp(u, 0); pd z1; pd z3; draw z1 .. z2 .. z3; % stroke
endchar;

beginpunkchar("[", 8, .5, .5);
z1 = pp(w - u, h); z2 = pp(.5w, h); z3 = pp(.5w, 0); z4 = pp(w - u, 0);
pd z1; pd z4; draw z1 -- z2 -- z3 -- z4; % bars and stem
endchar;

beginpunkchar("]", 8, .5, .5);
z1 = pp(u, h); z2 = pp(.5w, h); z3 = pp(.5w, 0); z4 = pp(u, 0);
pd z1; pd z4; draw z1 -- z2 -- z3 -- z4; % bars and stem
endchar;

beginpunkchar("<", 9, .5, .5);
z1 = pp(w - u, .9h); z2 = pp(u, .5h); z3 = pp(w - u, .1h);
pd z1; pd z3; draw z1 -- z2 -- z3; % diagonals
endchar;

beginpunkchar(">", 9, .5, .5);
z1 = pp(u, .9h); z2 = pp(w - u, .5h); z3 = pp(u, .1h);
pd z1; pd z3; draw z1 -- z2 -- z3; % diagonals
endchar;

beginpunkchar("=", 9, .5, .5);
z5 = pp(u, 2/3h); z6 = pp(w - u, 2/3h); pd z5; pd z6; draw z5 -- z6; % upper bar
z7 = pp(u, 1/3h); z8 = pp(w - u, 1/3h); pd z7; pd z8; draw z7 -- z8; % lower bar
endchar;

beginpunkchar("#", 15, .5, .5);
z1 = pp(.5w, h); z2 = pp(3u, 0); z3 = pp(w - 3u, h); z4 = pp(.5w, 0);
pd z2; pd z3; draw z3 -- z1 -- z2; draw z3 -- z4 -- z2; % diagonals (linked)
z5 = pp(u, 2/3h); z6 = pp(w - u, 2/3h); pd z5; pd z6; draw z5 -- z6; % upper bar
z7 = pp(u, 1/3h); z8 = pp(w - u, 1/3h); pd z7; pd z8; draw z7 -- z8; % lower bar
endchar;

```

```

beginpunkchar("/", 9, 1, 1);
z1 = pp(1.5u, -.05h); z2 = pp(w - 1.5u, 1.05h); pd z1; pd z2; draw z1 -- z2;           % diagonal
endchar;

beginpunkchar(oct "013", 12, .5, .5);
z1 = pp(u, .7h); z2 = pp(.5w, h); z3 = pp(w - u, .7h); z4 = pp(.5w, 0);
pd z1; pd z3; pd z4; draw z1 -- z2 -- z4; draw z3 -- z2;           % stem and diagonals
endchar;

beginpunkchar(oct "014", 12, .5, .5);
z1 = pp(u, .3h); z2 = pp(.5w, 0); z3 = pp(w - u, .3h); z4 = pp(.5w, h);
pd z1; pd z3; pd z4; draw z1 -- z2 -- z4; draw z3 -- z2;           % stem and diagonals
endchar;

```

The final program file, PUNKA.MF, defines accents in a form that T_EX likes. The T_EX input

```

\def\AA{\accent'27A}
{\AA}ngel\aa\ Beatrice Claire Diana \Erica Fran\c{c}oise
Ginette H\el\ene Iris Jackie K\=aren {\L}au\ra Mar{\'\i}a
N\H{a}ta{\l}{\u\i}e {\O}ctave Pauline Qu\eneau Roxanne Sabine
T\~a{\'\j}a Ur\v{s}ula Vivian Wendy Xanthippe Yv{\o}nne Z"azilie

```

causes accents to be positioned as follows, in the font PUNKSL20:

*ANGELA BEATRICE CLAIRE DIANA ÉRICA
FRANÇOISE GINETTE HÉLÈNE IRIS
JACKIE KÄREN LAURA MARÍA NATALIE
OCTAVE PAULINE QUÉNEAU ROXANNE
SABINE TĀĀJA ŪRŠŪLA VIVIAN WENDY
XANTHIPPE YVØNNE ZÄZILIE*

(Notice that the macro \AA needs to be redefined, but the other accents of plain T_EX work without change.)

Here is the way accents are drawn:

```

beginpunkchar(oct "022", 9, 1, 1);
z1 = pp(2.5u, h); z2 = pp(.6w, .8h); pd z1; pd z2; draw z1 -- z2;           % diagonal
endchar;

beginpunkchar(oct "023", 9, 1, 1);
z1 = pp(w - 2.5u, h); z2 = pp(.4w, .8h); pd z1; pd z2; draw z1 -- z2;           % diagonal
endchar;

beginpunkchar(oct "136", 13, 1, 1);
z1 = pp(2.5u, .8h); z2 = pp(.5w, h); z3 = (w - 2.5u, .8h);
pd z1; pd z3; draw z1 -- z2 -- z3;           % diagonals
endchar;

beginpunkchar(oct "024", 13, 1, 1);
z1 = pp(2.5u, .9h); z2 = pp(.5w, .7h); z3 = pp(w - 2.5u, .9h);
pd z1; pd z3; draw z1 -- z2 -- z3;           % diagonals
endchar;

```



```

beginpunkchar(oct "025", 11, 1, 1); % ~
z1 = pp(2u, h); z2 = pp(.5w, .75h); z3 = pp(w - 2u, h);
pd z1; pd z3; draw flex(z1, z2, z3); % stroke
endchar;

beginpunkchar(oct "026", 12, 1, 1); % -
z1 = pp(u, .8h); z2 = pp(w - u, .8h); pd z1; pd z2; draw z1 -- z2; % bar
endchar;

beginpunkchar(oct "137", 5, 1, 1); % ·
pd pp(.5w, .9h); % dot
endchar;

beginpunkchar(oct "177", 13, 1, 1); % ¨
pd pp(1/5w, .9h); pd pp(4/5w, .9h); % dots
endchar;

beginpunkchar(oct "176", 13, 1, 1); % ~
z1 = pp(u, .75h); z2 = pp(w - u, .9h); pd z1; pd z2; draw z1{up} .. {up}z2; % stroke
endchar;

beginpunkchar(oct "175", 13, 1, 1); % ¨
z1 = pp(4u, h); z2 = pp(2.5u, .7h); z3 = pp(w - 2u, h); z4 = pp(w - 3.5u, .7h);
pd z1; pd z3; draw z1 -- z2 -- z4 -- z3; % diagonals (linked)
endchar;

beginpunkchar(oct "027", 13, 0, 0); % Scandinavian loop, for Å and å
z0 = (.5w, .66h); % point z2 of lowercase Å
z1 = (.5w, .9h); draw z0{z0 - (1.5u, 0)} .. z1 .. {(w - 1.5u, 0) - z0}z0; % loop
endchar;

beginpunkchar(oct "030", 13, .5, .5); % Cedilla, for ç
z1 = (.6w, 0); z2 = pp(.6w, -.1h); z3 = pp(2.5u, -.1h);
pd z3; draw z1 -- z2 -- z3; % stroke
endchar;

beginpunkchar(oct "040", 11, .5, .5); % Polish cross, for Ł and ł
z1 = pp(0, .25h); z2 = pp(4u, .4h); pd z1; pd z2; draw z1 -- z2; % diagonal
endchar;
ligtable oct "040": "l" kern -charwd, "L" kern -charwd;

```

Bibliography

- [1] George Ade, *Artie: A Story of the Streets and Town*, 1896, Chapter 19.
- [2] Dashiell Hammett, *The Maltese Falcon*, 1930, Chapter 18.
- [3] Derek Jewell, music review in the *Sunday Times*, 28 November 1976, page 37.
- [4] Donald E. Knuth, *The METAFONT book*, Addison-Wesley, 1986, page v.
- [5] Pablo Picasso, from his sketchbooks. (This drawing was later used as an illustration in Vollard's de luxe edition of *Le Chef-d'Œuvre Inconnu* by Honoré de Balzac, 1931.)
- [6] William Shakespeare, *The Merry Wives of Windsor*, Act 2, Scene 2, lines 135–137. (The *First Folio* has the spelling 'Puncke'.)